



# Integrated Circuits Design by FPGA

م.م. أحمد مؤيد عبدالحسين  
جامعة الفرات الأوسط التقنية / الكلية التقنية الهندسية / نجف

# Lecture 10

## Packages and Components

# Objectives of this Lecture

- To define the new terms **Packages** and **Components**
- To understand the code structure of **Packages** and **Components**.
- To implement examples using **Packages** and **Components**.


# Contents of this Lecture

- Introduction
- PACKAGE
- COMPONENT

# Introduction

- Code structure: library declarations, entity, architecture (Lecture 1)
- VHDL Data Classes and Data Types (Lecture 2)
- VHDL Parallel Code (Lecture 4)
- VHDL Sequential Code (Lecture 5 & Lecture 6 )
- Design of Finite State Machines (FSM) (Lecture 8)

# Introduction



**Figure 10.1**  
Fundamental units of VHDL code.

# Introduction

- Packages



- Components

Lecture 10 & 11

- Functions



- Procedures

Lecture 12 & 13

# Introduction





- These new units can be located in the main code itself (that is, on the left-hand side of figure 10.1)
- However, since their main purpose is to allow common pieces of code to be reused and shared, it is more usual to place them in a **LIBRARY**.
- This also leads to code partitioning, which is helpful when dealing with long codes.
- In summary, frequently used pieces of code can be written in the form of **COMPONENTS**, **FUNCTIONS**, or **PROCEDURES**, then placed in a **PACKAGE**, which is finally compiled into the destination **LIBRARY**.
- We have already seen that at least two **LIBRARIES** are generally needed in a design: **ieee**, and **work**.
- After studying lectures 10 to 13, we will be able to construct our own libraries, which can then be added to the list above.



# Introduction

- For example, commonly used circuits, like flip-flops, multiplexers, adders, logic gates, etc., can be placed in a LIBRARY, so any project can make use of them without having to explicitly rewrite such codes.

```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.all;
3  -----
4  ENTITY and2 IS
5  PORT (a, b: IN std_logic_vector(3 downto 0);
6        x: OUT std_logic_vector(3 downto 0));
7  END and2;
8  -----
9  ARCHITECTURE and2 OF and2 IS
10 BEGIN
11   x <= a and b;
12 END and2;
13 -----
```



# PACKAGE

We start by describing the structure of a **PACKAGE**. Besides **COMPONENTS**, **FUNCTIONS**, and **PROCEDURES**, it can also contain **TYPE** and **CONSTANT** definitions, among others. Its syntax is presented below.

```
PACKAGE package_name IS  
    (declarations)  
END package_name;  
  
[PACKAGE BODY package_name IS  
    (FUNCTION and PROCEDURE descriptions)  
END package_name;]
```

As can be seen, the syntax is composed of two parts: PACKAGE and PACKAGE BODY

# PACKAGE

- The first part (PACKAGE) is mandatory and contains all declarations, while the second part (PACKAGE BODY) is necessary only when one or more subprograms (FUNCTION or PROCEDURE) are declared in the first upper part, in which case it must contain the descriptions (bodies) of the subprograms.
- PACKAGE and PACKAGE BODY must have the same name.
- The declarations list can contain the following: COMPONENT, FUNCTION, PROCEDURE, TYPE, CONSTANT, etc.

# PACKAGE

## Example 10.1: Simple Package

The example below shows a PACKAGE called my\_package. It contains only TYPE and CONSTANT declarations, so a PACKAGE BODY is not necessary.

```
1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  PACKAGE my_package IS
6      TYPE state IS (st1, st2, st3, st4);
7      TYPE color IS (red, green, blue);
8      CONSTANT vec: STD_LOGIC_VECTOR(7 DOWNT0 0) := "11111111";
9  END my_package;
10 -----
```

# PACKAGE

- The next example (example 10.2) contains, besides TYPE and CONSTANT declarations, a FUNCTION. Therefore, a PACKAGE BODY is now needed (details on how to write a FUNCTION will be seen in lecture 12). This function returns TRUE when a positive edge occurs on clk.

# PACKAGE

```
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 PACKAGE my_package IS
6     TYPE state IS (st1, st2, st3, st4);
7     TYPE color IS (red, green, blue);
8     CONSTANT vec: STD_LOGIC_VECTOR(7 DOWNT0 0) := "11111111";
9     FUNCTION positive_edge(SIGNAL s: STD_LOGIC) RETURN BOOLEAN;
10 END my_package;
11 -----
12 PACKAGE BODY my_package IS
13     FUNCTION positive_edge(SIGNAL s: STD_LOGIC) RETURN BOOLEAN IS
14     BEGIN
15         RETURN (s'EVENT AND s='1');
16     END positive_edge;
17 END my_package;
18 -----
```

# PACKAGE

Any of the PACKAGES above (example 10.1 or example 10.2) can now be compiled, becoming then part of our *work* LIBRARY (or any other). To make use of it in a VHDL code, we have to add a new USE clause to the main code (USE work.my\_package.all), as shown below.

```
-----  
LIBRARY ieee;  
USE ieee.std_logic_1164.all;  
USE work.my_package.all;  
-----  
ENTITY...  
...  
ARCHITECTURE...  
...  
-----
```

# COMPONENT

- A **COMPONENT** is simply a piece of conventional code (that is, **LIBRARY** declarations + **ENTITY** + **ARCHITECTURE**, as seen in previous lectures).
- However, by declaring such code as being a **COMPONENT**, it can then be used within another circuit, thus allowing the construction of hierarchical designs.
- A **COMPONENT** is also another way of partitioning a code and providing code sharing and code reuse.



# COMPONENT

- To use (instantiate) a COMPONENT, it must first be declared. The corresponding syntaxes are shown below.

COMPONENT declaration:

```
COMPONENT component_name IS
  PORT (
    port_name : signal_mode signal_type;
    port_name : signal_mode signal_type;
    ...);
END COMPONENT;
```

COMPONENT instantiation:

```
label: component_name PORT MAP (port_list);
```

# COMPONENT

----- COMPONENT declaration: -----

```
COMPONENT inverter IS
```

```
    PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
```

```
END COMPONENT;
```

----- COMPONENT instantiation: -----

```
U1: inverter PORT MAP (x, y);
```

# COMPONENT




Figure 10.2 a. : Declarations in the main code itself

# COMPONENT





Figure 10.2 b. : Declarations in a PACKAGE

# COMPONENT



**Figure 10.3**  
Circuit of example 10.3.

# Components Declared in the Main Code

```
1 ----- File inverter.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY inverter IS
6     PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
7 END inverter;
8 -----
9 ARCHITECTURE inverter OF inverter IS
10 BEGIN
11     b <= NOT a;
12 END inverter;
13 -----

1 ----- File nand_2.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY nand_2 IS
6     PORT (a, b: IN STD_LOGIC; c: OUT STD_LOGIC);
7 END nand_2;
8 -----
9 ARCHITECTURE nand_2 OF nand_2 IS
10 BEGIN
11     c <= NOT (a AND b);
12 END nand_2;
13 -----
```

```
1 ----- File nand_3.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY nand_3 IS
6     PORT (a, b, c: IN STD_LOGIC; d: OUT STD_LOGIC);
7 END nand_3;
8 -----
9 ARCHITECTURE nand_3 OF nand_3 IS
10 BEGIN
11     d <= NOT (a AND b AND c);
12 END nand_3;
13 -----

1 ----- File project.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY project IS
6     PORT (a, b, c, d: IN STD_LOGIC;
7         x, y: OUT STD_LOGIC);
8 END project;
9 -----
```



# Components Declared in the Main Code

```
10 ARCHITECTURE structural OF project IS
11     -----
12     COMPONENT inverter IS
13         PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
14     END COMPONENT;
15     -----
16     COMPONENT nand_2 IS
17         PORT (a, b: IN STD_LOGIC; c: OUT STD_LOGIC);
18     END COMPONENT;
19     -----
20     COMPONENT nand_3 IS
21         PORT (a, b, c: IN STD_LOGIC; d: OUT STD_LOGIC);
22     END COMPONENT;
23     -----
24     SIGNAL w: STD_LOGIC;
25 BEGIN
26     U1: inverter PORT MAP (b, w);
27     U2: nand_2 PORT MAP (a, b, x);
28     U3: nand_3 PORT MAP (w, c, d, y);
29 END structural;
30 -----
```

# Components Declared in a Package

```
1 ----- File inverter.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY inverter IS
6     PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
7 END inverter;
8 -----
9 ARCHITECTURE inverter OF inverter IS
10 BEGIN
11     b <= NOT a;
12 END inverter;
13 -----
```

```
1 ----- File nand_2.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY nand_2 IS
6     PORT (a, b: IN STD_LOGIC; c: OUT STD_LOGIC);
7 END nand_2;
8 -----
9 ARCHITECTURE nand_2 OF nand_2 IS
10 BEGIN
11     c <= NOT (a AND b);
12 END nand_2;
13 -----
```

```
1 ----- File nand_3.vhd: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY nand_3 IS
6     PORT (a, b, c: IN STD_LOGIC; d: OUT STD_LOGIC);
7 END nand_3;
8 -----
9 ARCHITECTURE nand_3 OF nand_3 IS
10 BEGIN
11     d <= NOT (a AND b AND c);
12 END nand_3;
13 -----
```



# Components Declared in a Package

```
1  ----- File my_components.vhd: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  PACKAGE my_components IS
6      ----- inverter: -----
7      COMPONENT inverter IS
8          PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
9      END COMPONENT;
10 ----- 2-input nand: ---
11 COMPONENT nand_2 IS
12     PORT (a, b: IN STD_LOGIC; c: OUT STD_LOGIC);
13 END COMPONENT;
14 ----- 3-input nand: ---
15 COMPONENT nand_3 IS
16     PORT (a, b, c: IN STD_LOGIC; d: OUT STD_LOGIC);
17 END COMPONENT;
18 -----
19 END my_components;
20 -----
```

```
1  ----- File project.vhd: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE work.my_components.all;
5  -----
6  ENTITY project IS
7      PORT ( a, b, c, d: IN STD_LOGIC;
8            x, y: OUT STD_LOGIC);
9  END project;
10 -----
11 ARCHITECTURE structural OF project IS
12     SIGNAL w: STD_LOGIC;
13 BEGIN
14     U1: inverter PORT MAP (b, w);
15     U2: nand_2 PORT MAP (a, b, x);
16     U3: nand_3 PORT MAP (w, c, d, y);
17 END structural;
18 -----
```

# Assignments

- The assignments will be attached to your class room.

**End of lecture 10**  
**Any Questions ?**